Disable Growl Notification When a Certain App is Running
Full credit to @ghoppe over at Super User for this excellent answer.  I'm only duplicating it here because a moderator on Meta suggested that it would be appropriate to do so.
Please note that this solution will suppress Growl when VLC (media player) is running.  To make it work for another application (e.g. QuickTime), you'll have to modify the Applescript.  If you've tried yourself but still need help, I would suggest posting a question about it over at Stack Overflow.
Original answer follows:


Enter in Applescript Editor, save as application, when saving check the box &quot;Stay Open&quot;. Use this new applescript application to launch the VLC application.
Description: It will launch VLC, turn off growl notifications, check every 2s to see if VLC has quit, if so it will turn growl notifications back on and then quit. As a bonus, it will use growl notifications to notify you when growl notifications will be turned on or off.

global Growl_was_Loaded    
global VLC_is_Loaded    

on run    
    set Growl_was_Loaded to isAppLoaded(&quot;GrowlHelperApp&quot;)    
    set VLC_is_Loaded to isAppLoaded(&quot;VLC&quot;)    
    launchVLC()    
    idle    
end run    

on idle    
    set x to isAppLoaded(&quot;VLC&quot;)    
    if x and not VLC_is_Loaded then    
        launchVLC()    
    else if VLC_is_Loaded and not x then    
        set VLC_is_Loaded to false    
        if Growl_was_Loaded then    
            tell application &quot;GrowlHelperApp&quot; to launch    
            growl_notify(&quot;Growl notifications have been turned ON&quot;)    
        end if    
        tell me to quit    
    end if    
    return 2 -- wait 2 seconds    
end idle    

on launchVLC()    
    tell application &quot;VLC&quot; to launch    
    if Growl_was_Loaded then    
        growl_notify(&quot;Launching VLC… Growl notifications have been turned OFF&quot;)    
        delay 1    
        tell application &quot;GrowlHelperApp&quot; to quit    
    end if    
    set VLC_is_Loaded to true    
end launchVLC    

on isAppLoaded(app_name)    
    tell application &quot;System Events&quot;    
        set app_list to every application process whose name is app_name    
        if the (count of app_list) &gt; 0 then    
            set x to true    
        else    
            set x to false    
        end if    
    end tell    
    return x    
end isAppLoaded    

on growl_notify(msg)    
    tell application &quot;GrowlHelperApp&quot;    
        set the allNotificationsList to {&quot;Growl Toggler&quot;}    
        register as application &quot;Growl Toggler&quot; all notifications allNotificationsList default notifications allNotificationsList    
        notify with name &quot;Growl Toggler&quot; title msg description &quot;&quot; application name &quot;Growl Toggler&quot; icon of application &quot;Automator&quot;    
    end tell    
end growl_notify